Zajecia 1

Krotkie wprowadzenie do srodowiska Mathematica

In[24]:= `2 + 2`

Out[24]= 4

In[25]:= `4^2`

Out[25]= 16

In[26]:= `4 × 4`

Out[26]= 16

In[27]:= `Sqrt[16]`

Out[27]= 4

In[28]:= `Log[10, 100]`

Out[28]= 2

In[29]:= `Sin[Pi / 2]`

Out[29]= 1

In[30]:= `4 !`

Out[30]= 24

In[31]:= `Factorial[4]`

Out[31]= 24

In[32]:= `Round[2.1]`

Out[32]= 2

In[33]:= `Mod[10, 3]`

Out[33]= 1

In[34]:= `? Mod`

Mod[*m*, *n*] gives the remainder on division of *m* by *n*.
Mod[*m*, *n*, *d*] uses an offset *d*. ≫

In[35]:= `E`

Out[35]= ⅇ

```
In[36]:= N[E]
Out[36]= 2.71828

In[37]:= N[Pi]
Out[37]= 3.14159

In[38]:= N[I]
Out[38]= 0. + 1. i

In[98]:= 3 / 9
Out[98]= 1/3

In[99]:= N[1 / 3]
Out[99]= 0.333333

In[41]:= %
Out[41]= 0.333333

In[100]:= %%
Out[100]= 1/3
```

Przypisywanie wartosci

```
In[102]:= x = 5
Out[102]= 5

In[112]:= x = y = 7;

In[104]:= Clear[x]
```

:= Assign the value to x, but don’ t do it right away; wait until x is actually used

```
In[105]:= x := 4

In[107]:= x == y
Out[107]= False

In[108]:= Solve[x^2 - 2 x + 1 == 0, x]
```

Solve::ivar : 4 is not a valid variable. ≫

```
Out[108]= Solve[False, 4]

In[109]:= Clear[x]

In[110]:= Solve[x^2 - 2 x + 1 == 0, x]
Out[110]= {{x → 1}, {x → 1}}

In[111]:= Clear[x]; Solve[x^2 - 7 x - 3 == 0, x]
```

Out[111]= $\left\{\left\{x \to \frac{1}{2}\left(7-\sqrt{61}\right)\right\}, \left\{x \to \frac{1}{2}\left(7+\sqrt{61}\right)\right\}\right\}$

Transformacje wyrazen algebraicznych

In[42]:=
```
x + x
```
Out[42]= 2 x

In[43]:=
```
(x + 1) ^2
```
Out[43]= $(1 + x)^2$

In[44]:=
```
Expand[(1 + x) ^2]
```
Out[44]= $1 + 2 x + x^2$

In[45]:=
```
Factor[%]
```
Out[45]= $(1 + x)^2$

Definiowanie funkcji

```
funkcja[x_] := x^2 + 1
```

In[47]:=
```
funkcja[2]
```
Out[47]= 5

Define a new function f.Note the underscore on the x on the left side of the
statement! It must be there; it tells Mathematica to treat x as a pattern; thereafter, when you type something like f[a + b], Mathematica will then immediately consider
that to be the same as (a + b)^3 - (a + b)

In[114]:=
```
funkcja2[x_, y_] := x^2 + y^2
```

In[115]:=
```
funkcja2[1, 2]
```
Out[115]= 5

https://mathematica.stackexchange.com/questions/19035/what-does-mean-in-mathematica

In[54]:=
```
? #
```

represents the first argument supplied to a pure function.
#*n* represents the n^{th} argument. ≫

In[55]:=
```
? &
```

Function[*body*] or *body* & is a pure function. The formal parameters are # (or #1), #2, etc.
Function[*x*, *body*] is a pure function with a single formal parameter *x*.
Function[{x_1, x_2, □ }, *body*] is a pure function with a list of formal parameters. ≫

In[56]:=
```
f := #^2 + 1 &
```

In[57]:=
```
f[2]
```
Out[57]= 5

In[116]:=
```
g := #1 ^2 + #2^3  + 10 &
```

In[117]:=
```
g[1, 2]
```
Out[117]= 19

Listy

In[66]:=
```
lista = {1, 2, 4, 8};
```

In[67]:=
```
lista[[3]]
```
Out[67]= 4

In[141]:=
```
? Table
```

Table[*expr*, {i_{max}}] generates a list of i_{max} copies of *expr*.
Table[*expr*, {*i*, i_{max}}] generates a list of the values of *expr* when *i* runs from 1 to i_{max}.
Table[*expr*, {*i*, i_{min}, i_{max}}] starts with $i = i_{min}$.
Table[*expr*, {*i*, i_{min}, i_{max}, *di*}] uses steps *di*.
Table[*expr*, {*i*, {i_1, i_2, … }}] uses the successive values i_1, i_2, ….
Table[*expr*, {*i*, i_{min}, i_{max}}, {*j*, j_{min}, j_{max}}, …] gives a nested list. The list associated with *i* is outermost. ≫

In[65]:=
```
Table[i^2, {i, 10}]
```
Out[65]= {1, 4, 9, 16, 25, 36, 49, 64, 81, 100}

In[137]:=
```
Table[i^2, {i, 2, 10, 2}]
```
Out[137]= {4, 16, 36, 64, 100}

In[68]:=
```
Table[Prime[i], {i, 50}]
```
Out[68]=
```
{2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67,
 71, 73, 79, 83, 89, 97, 101, 103, 107, 109, 113, 127, 131, 137, 139, 149,
 151, 157, 163, 167, 173, 179, 181, 191, 193, 197, 199, 211, 223, 227, 229}
```

In[139]:=
```
w := Table[i / j, {i, 4}, {j, 2}]
```

In[140]:=
```
w // MatrixForm
```
Out[140]//MatrixForm=

$$\begin{pmatrix} 1 & \frac{1}{2} \\ 2 & 1 \\ 3 & \frac{3}{2} \\ 4 & 2 \end{pmatrix}$$

In[121]:=
```
Table[If[EvenQ[i] || EvenQ[j], 1, 0] , {i, 3} , {j, 3}]
```
Out[121]= {{0, 1, 0}, {1, 1, 1}, {0, 1, 0}}

In[122]:=
```
% // MatrixForm
```
Out[122]//MatrixForm=

$$\begin{pmatrix} 0 & 1 & 0 \\ 1 & 1 & 1 \\ 0 & 1 & 0 \end{pmatrix}$$

Grafika

In[63]:=
```
Plot[Sin[x], {x, 0, 6 Pi}]
```

Out[63]=



In[64]:=
```
Manipulate[Plot[Sin[x (1 + a x)], {x, 0, 6}], {a, 0, 2}]
```

Out[64]=



In[120]:=
```
ListPlot[Table[Prime[i], {i, 50}]]
```

Out[120]=



In[70]:= `Graphics[Circle[]]`

Out[70]=



In[71]:= `Show[ListPlot[Range[10]], Plot[x + Sin[x], {x, 0, 10}]]`

Out[71]=



Litery greckiego alfabetu Esc+litera+Esc, np. θ, α.
Indeks dolny Ctrl + _

In[72]:= `? Rectangle`

Rectangle[{x_{min}, y_{min}}, {x_{max}, y_{max}}] is a two-dimensional graphics primitive that represents a filled rectangle, oriented parallel to the axes.
Rectangle[{x_{min}, y_{min}}] corresponds to a unit square with its bottom-left corner at {x_{min}, y_{min}}. ≫

In[73]:= `? Graphics`

Graphics[*primitives*, *options*] represents a two-dimensional graphical image. ≫

In[74]:=
```
Graphics[{Thick, Green, Rectangle[{0, -1}, {2, 1}], Red, Disk[],
  Blue, Circle[{2, 0}], Yellow, Polygon[{{2, 0}, {4, 1}, {4, -1}}]}]
```

Out[74]=



In[75]:=
```
? GraphicsGrid
```

GraphicsGrid[{{g_{11}, g_{12}, … }, … }] generates a graphic in which the g_{ij} are laid out in a two–dimensional grid. >>

In[76]:=
```
GraphicsGrid[{{Graphics[Rectangle[]], Graphics[Disk[]]},
  {Graphics[Disk[]], Graphics[Rectangle[]]}}]
```

Out[76]=



Obroty na okręgu

Zad. 1

In[77]:=
```
obrot[θ_, x_] := Mod[x + θ, 2 * Pi]
```

In[78]:=
```
obrot[Pi / 6, π / 6]
```

Out[78]= $\frac{\pi}{3}$

In[79]:=
```
obrot[θ_, x_] := Mod[N[x + θ], 2 * Pi]
```

In[80]:=
```
obrot[Pi / 6, 0]
```
Out[80]=
```
0.523599
```

Zad. 2

In[81]:=
```
Plot[obrot[Pi / 6, x], {x, 0, 6 Pi}, AxesLabel -> {"x", "fλ(x)"},
 PlotStyle → Red, AxesStyle → Arrowheads[0.05],
 LabelStyle → Directive[15, Black], PlotLabel → "Wykres funkcji obrotu"]
```

Out[81]=




```
Manipulate[ Plot[obrot[θ, x], {x, 0, 6 Pi}, AxesLabel -> {"x", "fθ(x)"},
  PlotStyle → Red, AxesStyle → Arrowheads[0.05],
  LabelStyle → Directive[15, Black], PlotLabel → "Wykres funkcji obrotu"],
 {{θ, Exp[1], "K„t obrotu" }, 0, 2 Pi}]
```

Out[82]=



Zad. 3

In[84]:=
```
Nest[h, x, 4]
```

Out[84]=
```
h[h[h[h[x]]]]
```

In[85]:=
```
NestList[f, x, 4]
```

Out[85]= $\{x, 1+x^2, 1+\left(1+x^2\right)^2, 1+\left(1+\left(1+x^2\right)^2\right)^2, 1+\left(1+\left(1+\left(1+x^2\right)^2\right)^2\right)^2\}$

In[86]:=
```
orbita[x0_, θ_, n_] := NestList[obrot[θ, #] &, x0, n - 1]
```

In[88]:=
```
orbita1[x0_, θ_, n_] := Table[Mod[x0 + i * θ, 2 Pi], {i, 0, n - 1, 1}]
```

If you want to group several commands and output the last use the semicolon (;) between them, Just don' t use a for the last statement.

In[89]:=
```
fun[y_] := (x = y + 5; x^2)
```

In[90]:=
```
fun[2]
```

Out[90]=
```
49
```

In[91]:=
```
Manipulate[zp = orbita[x0, θ, n];
 ListPlot[zp, PlotStyle → {Red, PointSize[Medium]}, PlotRange → {0, 2 Pi},
  AxesLabel → {"n", "fλ^(n) (x0)"}, LabelStyle → Directive[15, Black],
  PlotLabel → "wykres orbity"], {{θ, Exp[1], "K„t obrotu"}, 0, 2 Pi},
 {{x0, 2, "x0"},}, {{n, 40, "Max. d‡ugo□æ orbity"}, {40, 80, 160, 320}}]
```

Out[91]=



In[92]:=
```
Manipulate[zp = orbita[x0, θ, n];
  ListPlot[Table[{i - 1, zp[[i]]}, {i, 1, Length[zp]}],
   PlotStyle → {Red, PointSize[Medium]}, PlotRange → {0, 2 Pi},
   AxesLabel → {"n", "f_λ^(n) (x_0)"}, LabelStyle → Directive[15, Black],
   PlotLabel → "wykres orbity"], {{θ, Exp[1], "K„t obrotu"}, 0, 2 Pi},
  {{x0, 2, "x_0"},}, {{n, 40, "Max. d‡ugo□æ orbity"}, {40, 80, 160, 320}}]
```

Out[92]=



In[167]:=
```
Manipulate[zp = orbita[x0, θ, Max[n, k]]; Graphics[{Red, Circle[],
   Black, Table[Disk[{Sin[zp[[i]]], Cos[zp[[i]]]}, 0.02], {i, 1, k}]},
  PlotLabel → "wizualzacja orbity", LabelStyle → Directive[15, Black]],
 {{θ, Exp[1], "K„t obrotu"}, 0, 2 Pi}, {{x0, 2, "x0"}, 0, 2 Pi},
 {{n, 320, "Max. d‡ugo□æ orbity"}, {40, 80, 160, 320, 400}},
 {{k, n, "ilo□æ kroków"}, 1, n, 1}]
```

Out[167]=



In[94]:=
```
Histogram[RandomVariate[NormalDistribution[0, 1], 200]]
```

Out[94]=



In[159]:=
```
Histogram[orbita[Exp[1], Pi / 12, 15 000], {1 / 10},
 "PDF", ChartStyle → Green, PlotLabel → "Histogram orbity",
 LabelStyle → Directive[15, Black], PlotRange → {{0, 2 Pi}, All}]
```

Out[159]=



```
In[96]:= Manipulate[zp = orbita[x0, θ, Max[n, k]];
  GraphicsGrid[{
    {
     Graphics[{Red, Circle[], Black,
        Table[Disk[{Sin[zp[[i]]], Cos[zp[[i]]]}, 0.02], {i, 1, k}]},
      PlotLabel → "wizualzacja orbity",
      LabelStyle → Directive[15, Black]],

     ListPlot[Table[{i - 1, zp[[i]]}, {i, 1, k}],
      PlotStyle → {Red, PointSize[Medium]},
      PlotRange → {{0, n}, {0, 2 Pi}},
      AxesLabel → {"n", "fλ(n)(x0)"},
      LabelStyle → Directive[15, Black],
      PlotLabel → "wykres orbity"]
    },
    {
     Plot[obrot[θ, x], {x, 0, 4 Pi},
      AxesLabel → {"x", "fλ(x)"},
      PlotStyle → Red,
      AxesStyle → Arrowheads[0.05],
      LabelStyle → Directive[15, Black],
      PlotLabel → "Wykres funkcji obrotu",
      ImageSize → 50,
      Ticks → {{0, Pi, 2 Pi, 3 Pi, 4 Pi}, {0, Pi / 2, Pi, 3 Pi / 2}},
      AxesOrigin → {0, 0}],

     Histogram[zp, {1 / 10}, "PDF",
      ChartStyle → Green,
      PlotLabel → "Histogram orbity",
      LabelStyle → Directive[15, Black],
      PlotRange → {{0, 2 Pi}, All}]
    }
   }
   , ImageSize → 650],
  {{θ, Exp[1], "K„t obrotu"}, 0, 2 Pi},
  {{x0, 2, "x0"}, 0, 2 Pi},
  {{n, 40, "Max. d‡ugo□æ orbity"},
   {40, 80, 160, 320, 640}},
  {{k, n, "ilo□æ kroków"}, 1, n, 1}]
```

Out[96]=

Kąt obrotu

x_0

Max. długość orbity | 40 | 80 | 160 | 320 | 640

ilość kroków